

NASA TM X-1339

**COMPUTER PROGRAM FOR THERMODYNAMIC PERFORMANCE
OF BRAYTON-CYCLE SPACE-POWER SYSTEMS**

By Arthur J. Glassman

**Lewis Research Center
Cleveland, Ohio**

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

**For sale by the Clearinghouse for Federal Scientific and Technical Information
Springfield, Virginia 22151 - Price \$1.00**

**BLANK
PAGE**

COMPUTER PROGRAM FOR THERMODYNAMIC PERFORMANCE OF BRAYTON-CYCLE SPACE-POWER SYSTEMS

**by Arthur J. Glassman
Lewis Research Center**

SUMMARY

A computer program used for the calculation of the thermodynamic performance of one- and two-shaft Brayton-cycle space-power systems is presented. Systems that can be analyzed include those with and without reheating and/or intercooling. Provision is made in the program to account for turbine-coolant flow.

Inputs required for the program include the component performance parameters and the cycle temperature variables. Performance outputs from the program include cycle efficiency and prime radiator area. Sample input and output are presented to demonstrate the use of the program.

INTRODUCTION

One of the first phases in the analysis of any power cycle is the determination of the thermodynamic performance parameters and optimum cycle-variable values pertinent to the particular application. Such analyses have been performed for Brayton-cycle space-power systems to determine the effects of cycle temperature variables, component performance parameters, reheating and/or intercooling, and turbine-coolant bypass flow on cycle efficiency and prime radiator area. Examples of these analyses are found in references 1 and 2. These cycle studies are performed most rapidly and conveniently with a computer.

The computer program used to obtain the thermodynamic performance results in references 1 and 2 and associated studies is presented herein. The analysis equations comprising the program are given in these references. One- and two-shaft systems with and without reheating and/or intercooling can be analyzed. It is also possible to account for the effects of turbine-coolant flow. The features of this program are discussed and

the FORTRAN listing is presented. Sample input and output are included to demonstrate use of the program.

PROGRAM DESCRIPTION

The computer program calculates cycle thermodynamic performance for one- and two-shaft Brayton-cycle space-power systems with and without reheating and/or intercooling. A schematic diagram of the general system being analyzed along with calculation station identification numbers is presented in figure 1. The system variations that can be analyzed are obtained by eliminating components such as the reheater, intercooler, and high-pressure turbomachinery through input specifications. The thermodynamic analysis equations that comprise the program were derived in references 1 and 2. The general features of the program are discussed herein and detailed descriptions of the input and output are presented. The FORTRAN program and associated variables are listed in the appendixes.

General Features

Basic systems. - The program is set up to analyze five basic systems, four of which are included in the general system shown in figure 1. These four systems, in which the




Figure 1. - Brayton-cycle system with intercooling and reheating.

alternator is on the low-pressure shaft, include those with (1) neither reheating nor intercooling, (2) reheating only, (3) intercooling only, and (4) both reheating and intercooling. In the fifth system, which also has both reheating and intercooling, the alternator is on the high-pressure shaft. As seen in figure 1, only one stage of reheating or intercooling is considered. The system to be analyzed is specified by the input variable KASE, as will be described in the section Input, and the program automatically selects the proper analysis path.

Shaft arrangements. - As seen in figure 1, the system is considered to have a two-shaft turbomachinery arrangement when reheating and/or intercooling are used. With neither reheating nor intercooling, either one-shaft or two-shaft arrangements can be analyzed. The selected shaft arrangement and particular turbine-work split for a two-shaft arrangement are specified by the input variable ST, which expresses the ratio of high-pressure-turbine work to low-pressure-turbine work. Appropriate selection of ST, as will be discussed in the section Input, yields shaft arrangements including a one-shaft system (the low-pressure shaft in figure 1), various two-shaft two-compressor systems (representing variations in shaft-work split), and a two-shaft one-compressor system (compressor on the high-pressure shaft).

Turbine-coolant flow. - The effects of turbine-coolant flow are analyzed according to the bypass model specified in reference 2; that is, coolant flow is assumed to bypass the turbine and then mix with the main stream. Coolant flow, as seen from figure 1, can originate either from the compressor outlet (station 5) or the recuperator outlet (station 6) as desired. This coolant can then rejoin the main stream downstream of either turbine. The amount, source, and destination of the bypass flow are specified by the input variables Y1P, Y1PFR6, Y2, and Y2FR6 as described in the section Input.

Temperature ratio variation. - For each set of input parameters, the two cycle temperature variables can be varied parametrically between specified limits. These variables are cycle temperature ratio (T41), which is the ratio of compressor-inlet to turbine-inlet temperature, and turbine temperature ratio (T21), which is the ratio of turbine-exit to turbine-inlet temperature. The limits and increments for these temperature ratios are specified by the input variables T41MIN, T41DEL, T41MAX, T21MIN, T21DEL, and T21MAX. For each value of cycle temperature ratio, the cycle computations are made for the range of turbine temperature ratios, and then the maximum cycle efficiency within that range and the associated turbine temperature ratio are determined.

Reheat and intercool temperatures. - For any given case, the temperatures leaving the intercooler or reheater can be varied. These temperatures are expressed as ratios of the first-compressor-inlet or first-turbine-inlet temperatures, and are specified by the input variables C4 and C1.

Program organization. - There is a main program, ETACY2, and three subroutines, RADTRA, MAXIM, and ZERO. Main program ETACY2 controls all the input and output

and computes all temperatures and turbomachinery pressure ratios, as well as cycle efficiency and weight flow, for the cycle. Subroutine RADTRA computes the prime radiator areas required for both primary heat rejection and intercooling. Subroutine MAXIM provides the logic for determining the maximum cycle efficiency and associated turbine temperature ratio for each cycle temperature ratio. Subroutine ZERO provides the logic for determining the turbine work split that yields a two-shaft one-compressor arrangement. These programs are listed and the variables identified in the appendixes.

Input

The input values for this program include component efficiencies, cycle loss pressure ratio, gas specific heat ratio, turbine work split, turbine-inlet temperature, space-sink temperature, radiator emissivity, gas heat-transfer coefficient in the radiator, reheat temperature, intercool temperature, cycle temperature-ratio range and increment, turbine temperature-ratio range and increment, and integers specifying the appropriate system and giving instructions for further input.

In table I, sample data are presented in the required form for input. A field width of 10 columns is allowed for all variables except those on the first card, which contains only 2 one-digit integers. If there is no decimal point included in the field, the four rightmost elements of each field are taken to represent four decimal places. The input variables are defined in the following list:

KASE	specifies system as follows: <ul style="list-style-type: none">1 - intercooling and reheating, alternator on high-pressure shaft2 - intercooling only, alternator on low-pressure shaft3 - intercooling and reheating, alternator on low-pressure shaft4 - no intercooling or reheating, alternator on low-pressure shaft5 - reheating only, alternator on low-pressure shaft
KREAD	specifies FORTRAN statement number (1, 2, or 3) to which program returns for more input data
ETAT1	efficiency of turbine on high-pressure shaft, fraction
ETAT2	efficiency of turbine on low-pressure shaft, fraction
ETAC1	efficiency of compressor on low-pressure shaft, fraction
ETAC2	efficiency of compressor on high-pressure shaft, fraction
RL	cycle loss pressure ratio
E	recuperator effectiveness, fraction

GAMMA	gas specific heat ratio
T41MIN	minimum value of cycle temperature ratio (ratio of compressor-inlet to turbine-inlet temperature)
T41DEL	cycle temperature-ratio increment
T41MAX	maximum value of cycle temperature ratio
T21MIN	minimum value of turbine temperature ratio (ratio of turbine-exit to turbine-inlet temperature)
T21DEL	turbine temperature-ratio increment
T21MAX	maximum value of turbine temperature ratio
C1	ratio of reheat temperature to turbine-inlet temperature
C4	ratio of intercool temperature to compressor-inlet temperature
ST	ratio of high-pressure-turbine work to low-pressure-turbine work (ST = 0 yields a one-shaft arrangement and a negative ST yields a two-shaft, one-compressor arrangement)
T1	turbine-inlet temperature, $^{\circ}\text{R}$
TS	space-sink temperature, $^{\circ}\text{R}$
EPS	radiator surface emissivity
HR	radiator gas heat-transfer coefficient, $\text{Btu}/(\text{hr})(\text{ft}^2 \text{ prime area})(^{\circ}\text{R})$
DLMIN	convergence increment for radiator area computation
Y1P	ratio of high-pressure-turbine coolant flow to compressor flow
Y1PFR6	fraction of Y1P originating from recuperator outlet
Y2	ratio of low-pressure-turbine coolant flow to compressor flow
Y2FR6	fraction of Y2 originating from recuperator outlet
DLT1PM	convergence increment for temperature after mixing of main stream and coolant

Output

The printed output from the program includes pertinent input values and the following computed values: cycle efficiency, prime radiator area, a weight flow parameter, turbo-machinery pressure ratios, and ratios of all temperatures around the cycle to turbine-

TABLE II. - OUTPUT FORM WITH SAMPLE DATA

**BRAYTON CYCLE - TWO SPOOL - VARIABLE WORK SPLIT
NO INTERCOOL OR REHEAT - ALTERNATOR ON LOW PRESSURE SHAFT**

ETA(T1)	ETA(T2)	ETA(C1)	ETA(C2)	R(L)	E	GAMMA	C1	C4	SPLIT(T)	AIN(T/P	ARAD/P	ATOT/P
T1	TS	EPS	HR		YIP	YIPFR6	Y2	Y2FR6				
3000.00	400.60	0.86	50.00	0.8700	0.	0.	0.0800	1.0000	0.			
T4/T1=0.2750 T4PP/T1=0. T1PP/T1=0.												
T2/T1	ETAC1	ETAC2	T3/T1	T4P/T1	T5/T1	T6/T1	P1P/P1P	P1PP/P2	P4P/P4	P5/P4PP	SPLIT(C)	AIN(T/P
0.6000	0.1676	15.048	1.0000	0.5800	0.5800	0.5910	1.0000	4.4442	5.1082	1.0000	0.	7.68
0.6500	0.2025	12.489	0.5742	0.5210	0.5210	0.5922	1.0000	3.4860	4.0069	1.0000	0.	6.31
0.7000	0.2133	12.026	1.0000	0.4721	0.4721	0.5979	1.0000	2.7939	3.2114	1.0000	0.	6.03
0.7500	0.2052	12.790	1.0000	0.4308	0.4308	0.6070	1.0000	2.2799	2.6206	1.0000	0.	6.39
0.8000	0.1810	14.946	1.0000	0.3955	0.3955	0.6188	1.0000	1.8892	2.1715	1.0000	0.	7.47
0.8500	0.1421	19.763	1.0000	0.3650	0.3650	0.6327	1.0000	1.5861	1.8231	1.0000	0.	9.90
0.9000	0.0886	33.089	1.0000	0.3383	0.3383	0.6484	1.0000	1.3470	1.5483	1.0000	0.	16.65
0.7020	0.2133	12.034	1.0000	0.4704	0.4704	0.5982	1.0000	2.7709	3.1850	1.0000	0.	6.03
T4/T1=0.3000 T4PP/T1=0. T1PP/T1=0.												
T2/T1	ETAC1	ETAC2	T3/T1	T4P/T1	T5/T1	T6/T1	P1P/P1P	P1PP/P2	P4P/P4	P5/P4PP	SPLIT(C)	AIN(T/P
0.6000	0.0996	26.858	1.0000	0.6327	0.6327	0.6146	1.0000	4.4442	5.1082	1.0000	0.	10.69
0.6500	0.1507	17.702	1.0000	0.5684	0.5684	0.6134	1.0000	3.4860	4.0069	1.0000	0.	6.94
0.7000	0.1731	15.540	1.0000	0.5150	0.5150	0.6171	1.0000	2.7939	3.2114	1.0000	0.	6.04
0.7500	0.1737	15.810	1.0000	0.4700	0.4700	0.6246	1.0000	2.2799	2.6206	1.0000	0.	6.10
0.8000	0.1564	18.064	1.0000	0.4315	0.4315	0.6349	1.0000	1.8892	2.1715	1.0000	0.	6.95
0.8500	0.1228	23.823	1.0000	0.3982	0.3982	0.6476	1.0000	1.5861	1.8231	1.0000	0.	9.18
0.9000	0.0737	41.403	1.0000	0.3691	0.3691	0.6622	1.0000	1.3470	1.5483	1.0000	0.	16.01
0.7254	0.1758	15.452	1.0000	0.4912	0.4912	0.6205	1.0000	2.5147	2.8904	1.0000	0.	5.97
T4/T1=0.3250 T4PP/T1=0. T1PP/T1=0.												
T2/T1	ETAC1	ETAC2	T3/T1	T4P/T1	T5/T1	T6/T1	P1P/P1P	P1PP/P2	P4P/P4	P5/P4PP	SPLIT(C)	AIN(T/P
0.6000	0.0228	124.840	1.0000	0.6502	0.6502	0.6383	1.0000	4.4442	5.1082	1.0000	0.	39.75
0.6500	0.0929	30.382	1.0000	0.6299	0.6299	0.6347	1.0000	3.4860	4.0069	1.0000	0.	9.49
0.7000	0.1257	22.036	1.0000	0.6165	0.6165	0.6364	1.0000	2.7939	3.2114	1.0000	0.	6.78
0.7500	0.1392	20.695	1.0000	0.6084	0.6084	0.6421	1.0000	2.2799	2.6206	1.0000	0.	6.31
0.8000	0.1294	22.827	1.0000	0.6045	0.6045	0.6510	1.0000	1.8892	2.1715	1.0000	0.	6.93
0.8500	0.1019	29.982	1.0000	0.6039	0.6039	0.6624	1.0000	1.5861	1.8231	1.0000	0.	9.10
0.9000	0.0575	55.297	1.0000	0.6060	0.6060	0.6759	1.0000	1.3470	1.5483	1.0000	0.	16.82
0.7488	0.1392	20.685	1.0000	0.6086	0.6086	0.6419	1.0000	2.2904	2.6326	1.0000	0.	6.31

inlet temperature. All temperatures and pressures are stagnation values. Prime radiator areas are computed on the basis that the gas cooler and intercooler are radiators.

Sample output corresponding to the sample input is given in table II. The top line is a program identification title; the second line specifies the case being analyzed. Next are two sets of two lines each, consisting of pertinent input variables and their associated values. The printed names for all but one of these input variables, except for added parentheses in some cases, corresponds exactly to those listed in the section Input. The one exception is SPLIT (T) which corresponds to the input variable ST. The computed results are then presented in groups, each group being for a different cycle temperature ratio (T_4/T_1) starting from the lowest (T_{41MIN}) and going to the highest (T_{41MAX}). Three such groups are shown in table II for cycle temperature ratios of 0.275, 0.300, and 0.325. The first line in each group identifies the cycle temperature ratio (T_4/T_1) and gives the ratios of intercool temperature to turbine-inlet temperature (T_{4PP}/T_1 , which is a function of the input variable C4) and reheat temperature to turbine inlet temperature (T_{1PP}/T_1 , which corresponds to the input variable C1). With no intercooling or reheating, the pertinent temperature ratio is artificially set to zero for printout purposes only. Next is a line of headings which are identified in the following list. Then there is one line of results for each turbine temperature ratio (T_2/T_1). The last line of each group shows the values that correspond to maximum cycle efficiency for that cycle temperature ratio group. The column headings for each group are identified as follows:

T_2/T_1	ratio of turbine-exit to turbine-inlet temperature
ETACY	cycle efficiency, fraction
WCT/P	weight flow parameter, equal to weight flow (lb/sec) times specific heat (Btu/(lb)($^{\circ}$ R)) times turbine-inlet temperature ($^{\circ}$ R) divided by shaft power to alternator (kW)
T_{1P}/T_1	ratio of high-pressure-turbine-exit to turbine-inlet temperature
T_3/T_1	ratio of gas cooler-inlet to turbine-inlet temperature
T_{4P}/T_1	ratio of low-pressure-compressor-exit to turbine-inlet temperature
T_5/T_1	ratio of high-pressure-compressor-exit to turbine-inlet temperature
T_6/T_1	ratio of gas-heater-inlet to turbine-inlet temperature
P_1/P_{1P}	pressure ratio across high-pressure turbine
P_{1PP}/P_2	pressure ratio across low-pressure turbine
P_{4P}/P_4	pressure ratio across low-pressure compressor
P_5/P_{4PP}	pressure ratio across high-pressure compressor

SPLIT (C) ratio of low-pressure-compressor work to high-pressure-compressor work
 (artificially set to zero when $ST = 0$)

ARAD/P specific prime radiator area required for primary heat rejection, ft^2/kW

AINT/P specific prime radiator area required for intercooling, ft^2/kW

ATOT/P total specific prime radiator area, ft^2/kW

Lewis Research Center,
 National Aeronautics and Space Administration,
 Cleveland, Ohio, November 15, 1966,
 128-31-02-25-22.

APPENDIX A

MAIN PROGRAM ETACY2

Main program ETACY2 controls all input and output, contains all analysis-path branching logic, and computes all temperature ratios around the cycle, turbomachinery pressure ratios, cycle efficiency, and weight flow parameter. The thermodynamic equations used are those presented in appendixes B of references 1 and 2.

PROGRAM VARIABLES

All items marked input variable are identified in the section Input. The variables for ETACY2 are the following:

AINOP	specific prime radiator area for intercooling
AROP	specific prime radiator area for primary heat rejection
ATOP	total specific prime radiator area
C1	input variable
C4	input variable
CFUNC	function of COEFC
COEFB	coefficient B of quadratic equation $x^2 + Bx + C = 0$ where x equals T4P1
COEFC	coefficient C of quadratic equation $x^2 + Bx + C = 0$ where x equals T4P1
DLMIN	input variable
DLT1PM	input variable
E	input variable
EE	product of recuperator effectiveness and fraction of compressor flow that passes through recuperator
EPS	input variable
ETACY	cycle efficiency
ETAC1	input variable
ETAC2	input variable
ETANUM	numerator of expression used to compute ETACY

ETAT1	input variable
ETAT2	input variable
EXP	function of GAMMA
GAMMA	input variable
HR	input variable
IN	logic indicator for determination of ST value that yields $SC = 0$
IND	logic indicator for determination of maximum value of ETACY
J	logic indicator for determination of program path during ETACY maximization iteration
KASE	input variable
KREAD	input variable
M	logic indicator for determination of T1PM1
P1PP2	pressure ratio across low-pressure turbine
P1P1X	isentropic temperature ratio across high-pressure turbine
P11P	pressure ratio across high-pressure turbine
P21PPX	isentropic temperature ratio across low-pressure turbine
P4P4	pressure ratio across low-pressure compressor
P4P4X	isentropic temperature ratio across low-pressure compressor
P54PP	pressure ratio across high-pressure compressor
P54PPX	isentropic temperature ratio across high-pressure compressor
RCX	isentropic temperature ratio corresponding to total pressure ratio across both compressors
RL	input variable
RTX	isentropic temperature ratio corresponding to total pressure ratio across both turbines
SC	ratio of low-pressure-compressor work to high-pressure-compressor work
ST	input variable
STT	input value of ST
STY	value of ST adjusted for coolant bypass
TEST	value tested for convergence of T1PM1

TS	input variable
TS1	ratio of TS to T1
TT1PM1	temporary storage for T1PM1
T1	input variable
T1PM1	ratio of high-pressure-turbine-exit temperature, after mixing with coolant, to turbine-inlet temperature
T1PP1	ratio of reheater-exit to turbine-inlet temperature
T1P1	ratio of high-pressure-turbine-exit to turbine-inlet temperature
T2M1	ratio of low-pressure-turbine-exit temperature, after mixing with coolant, to turbine-inlet-temperature
T21	ratio of low-pressure-turbine-exit to turbine-inlet temperature
T21BEG	lowest value of T21 used in ETACY maximization iteration
T21DEL	input variable
T21END	highest value of T21 used in ETACY maximization iteration
T21MAX	input variable
T21MIN	input variable
T31	ratio of gas cooler-inlet to turbine-inlet temperature
T4PP1	ratio of intercooler-exit to turbine-inlet temperature
T4P1	ratio of intercooler-inlet to turbine-inlet temperature
T4P4	temperature ratio across low-pressure compressor
T41	ratio of compressor-inlet to turbine-inlet temperature
T41DEL	input variable
T41MAX	input variable
T41MIN	input variable
T51	ratio of high-pressure-compressor-exit to turbine-inlet temperature
T54PP	temperature ratio across high-pressure compressor
T61	ratio of gas-heater-inlet to turbine-inlet temperature
WPARAM	weight flow parameter (see section Output)
XT1P1	storage for calculated T1P1
Y	lower limit test value for T1P1

YTOT ratio of total coolant to compressor flow
 Y1P input variable
 Y1PFR6 input variable
 Y2 input variable
 Y2FR6 input variable
 Y5 ratio of coolant flow from compressor exit to compressor flow
 Y51P ratio of high-pressure-turbine coolant flow from compressor exit to compressor flow
 Y52 ratio of low-pressure-turbine coolant flow from compressor exit to compressor flow
 Y6 ratio of coolant flow from recuperator exit to compressor flow
 Y61P ratio of high-pressure-turbine coolant flow from recuperator exit to compressor flow
 Y62 ratio of low-pressure-turbine coolant flow from recuperator exit to compressor flow

PROGRAM LISTING

\$IBFTC ETACY2

```

C      BRAYTON CYCLE - TWO SPOOL - VARIABLE WORK SPLIT
C      CASE 1 - INTERCOOL AND REHEAT - ALT. ON HIGH PRESSURE SHAFT
C      CASE 2 - INTERCOOL ONLY - ALT. ON LOW PRESSURE SHAFT
C      CASE 3 - INTERCOOL AND REHEAT - ALT. ON LOW PRESSURE SHAFT
C      CASE 4 - NO INTERCOOL OR REHEAT - ALT. ON LOW PRESSURE SHAFT
C      CASE 5 - REHEAT ONLY - ALT. ON LOW PRESSURE SHAFT
C      KASE DENOTES THE SPECIFIED CASE
C      END OF PROGRAM TRANSFER TO STATEMENT NUMBER KREAD
C      IF INPUT ST IS NEGATIVE, SC WILL EQUAL ZERO
C
1 READ (5,999) KASE,KREAD
  READ (5,1000) ETAT1,ETAT2,ETAC1,ETAC2
  READ (5,1000) RL,E,GAMMA
  READ (5,1000) T41MIN,T41DEL,T41MAX
  READ (5,1000) T21MIN,T21DEL,T21MAX
  T41MAX=T41MAX+T41DEL*.1
  T21MAX=T21MAX+T21DEL*.1
2 READ (5,1000) C1,C4,ST
  IF(C1.EQ.0.0) GO TO 1
  STT=ST
  READ (5,1000) T1,TS,EPS,HR,DLMIN
3 READ (5,1000) Y1P,Y1PFR6,Y2,Y2FR6,DLT1PM
  IF(Y1P.EQ.1.0) GO TO 2
  
```

```

15 WRITE (6,1010)
   XT1P1=1.0
   GO TO (30,20,40,41,42),KASE
20 WRITE (6,1021)
   GO TO 50
30 WRITE (6,1022)
   GO TO 50
40 WRITE (6,1023)
   GO TO 50
41 WRITE (6,1024)
   GO TO 50
42 WRITE (6,1025)
50 WRITE (6,1030) ETAT1,ETAT2,ETAC1,ETAC2,RL,E,GAMMA,C1,C4,ST
   WRITE (6,1035) T1,TS,EPS,HR,Y1P,Y1PFR6,Y2,Y2FR6
   Y62=Y2FR6*Y2
   Y61P=Y1PFR6*Y1P
   Y6=Y62+Y61P
   YTOT=Y2+Y1P
   Y5=YTOT-Y6
   Y52=Y2-Y62
   Y51P=Y1P-Y61P
   EE=E*(1.0-Y5)
   T41=T41MIN
   T21=T21MIN
   T51=T21
   T61=T21
   TT1PM1=0.0
   J=0
   M=0
55 T21BEG=T21MIN
   T21END=T21MAX
   T4PP1=C4*T41
   T1PP1=C1
   IF(KASE.EQ.2.OR.KASE.EQ.4) T1PP1=0.0
   IF(KASE.EQ.4.OR.KASE.EQ.5) T4PP1=0.0
   WRITE (6,1045) T41,T4PP1,T1PP1
49 IF(STT.LT.-.001) ST=1.0
   IN=1
51 IF(KASE.EQ.1.OR.KASE.EQ.3.OR.KASE.EQ.5) GO TO 60
   STY=ST*(1.0-Y2)/(1.0-YTOT)
   T1PM1=((1.0-YTOT)*(1.0+STY*T21)+Y51P*T51+Y61P*T61)/((1.0-Y2)*(1.0+ST))
   IF(T1PM1.LT.T21) T1PM1=T21
   T1P1=1.-STY*(T1PM1-T21)
   T1PP1=T1PM1
   IF(Y1P.EQ.0.0) TT1PM1=T1PM1
   TEST=ABS(T1PM1-TT1PM1)
   IF(TEST.LE.DLT1PM) M=1
   TT1PM1=T1PM1
   GO TO 70
60 T1PP1=C1
   M=1
   STY=ST*(1.0-Y2)/(1.0-YTOT)
   T1P1=1.-STY*(T1PP1-T21)
   XT1P1=T1P1
   Y=1.-ETAT1
   IF(T1P1.LE.Y) T1P1=Y+.01

```



```

70 T4PP1=C4*T41
   P1P1X=1.-(1.-T1P1)/ETAT1
   P21PPX=1.-(1.-T21/T1PP1)/ETAT2
   RTX=1./(P1P1X*P21PPX)
   EXP=1/GAMMA-1./GAMMA
   RCX=RTX/PL**EXP
   IF(KASE.EQ.1) GO TO 80
   IF(KASE.EQ.4.OR.KASE.EQ.5) GO TO 85
75 T51=(1.-YTOT)*(1.-T1P1)+T4PP1
   P54PPX=1.+(T51/T4PP1-1.)*ETAC2
   P4P4X=RCX/P54PPX
   IF(KASE.GE.4) GO TO 90
   T4P4=1.+(P4P4X-1.)/ETAC1
   T4P1=T4P4*T41
   GO TO 90
80 T4P1=(1.-Y2)*(T1PP1-T21)+T41
   P4P4X=1.+(T4P1/T41-1.)*ETAC1
   P54PPX=RCX/P4P4X
   T54PP=1.+(P54PPX-1.)/ETAC2
   T51=T54PP*T4PP1
   GO TO 90
85 COEFB=(1.-ETAC1-RCX+ETAC1*ETAC2*(1.-YTOT)*(1.-T1P1)/T41)*T41/ETAC1
   COEFC=T41*(1.-ETAC1)*ETAC2*(1.-YTOT)*(1.-T1P1)/ETAC1
   CFUNC=-SQRT(4.*COEFC)
   IF(COEFB.GT.CFUNC) GO TO 400
   T4P1=(-COEFB+SQRT(COEFB**2-4.*COEFC))/2.
   T4PP1=T4P1
   GO TO 75
90 T2M1=((1.-Y2)*T21+Y52*T51+Y62*(1.-E)*T51)/(1.-Y62*E)
   T31=EE*T51+(1.-EE)*T2M1
   T61=E*T2M1+(1.-E)*T51
   IF(M.EQ.0) GO TO 51
   M=0
   TT1PM1=0.0
   IF(KASE.EQ.1.OR.KASE.EQ.3.OR.KASE.EQ.5) T1PM1=(1.-YTOT)*T1P1+Y51P*
1T51+Y61P*T61
   IF(T21.EQ.1.0) GO TO 94
   IF(T1P1.EQ.1.0) GO TO 900
   SC=(T4P1-T41)/(T51-T4PP1)
   IF(STT.LT.-.001) CALL ZERO (2.0,ST,SC,IN,.0002)
   IF(IN.EQ.4) GO TO 400
   IF(STT.LT.-.001.AND.IN.NE.3) GO TO 51
94 ETANUM=(1.-YTOT)*(1.-T1P1)+(1.-Y2)*(T1PP1-T21)-T4P1+T41-T51+T4PP1
95 ETACY=ETANUM/((1.-YTOT)*(1.-T61)+(1.-Y2)*(T1PP1-T1PM1))
   IF(ETANUM.NE.0.0) GO TO 96
   WPARAM=0.0
   GO TO 97
900 SC=0.0
   GO TO 94
96 WPARAM=.9487/ETANUM
97 P11P=P1P1X**(-1./EXP)
   P1PP2=P21PPX**(-1./EXP)
   P4P4=P4P4X**(-1./EXP)
   P54PP=P54PPX**(-1./EXP)
   IF(J.NE.0) GO TO 100

```

```

190 IF(J.NE.0) WRITE (6,1060)
    TS1=TS/T1
    IF(SC.LT.0.0.OR.ETACY.LE.0.0.OR.T41.LE.TS1.OR.T4PP1.LE.TS1)GO TO 5
    CALL RADTRA(T1,TS,T31,T41,T4P1,T4PP1,EPS,HR,WPARAM,DLMIN,AINOP,
    1AROP,ATOP)
200 WRITE(6,1065)T21,ETACY,WPARAM,T1P1,T31,T4P1,T51,T61,P11P,P1PP2,P4P
    14,P54PP,SC,AROP,AINOP,ATOP
    GO TO 180
    5 WRITE(6,1070)T21,ETACY,WPARAM,T1P1,T31,T4P1,T51,T61,P11P,P1PP2,P4P
    14,P54PP,SC
180 IF(J.NE.0) GO TO 230
98 T21=T21+T21DEL
    IF(XT1P1.LE.Y) T21BEG=T21
    IF(ETACY.LT.0.0.AND.T21-T21BEG.LT.1.1*T21DEL) T21BEG=T21
    IF(ETACY.GE.0.0) T21END=T21-T21DEL
    IF(T21.LT.T21MAX) GO TO 49
    J=1
    T21=T21BEG
    IND=1
    GO TO 49
100 CONTINUE
    CALL MAXIM (T21,T21END,ETACY,IND,.005)
    IF(IND.NE.6) GO TO 49
    GO TO 190
230 J=0
    T21=T21MIN
    T41=T41+T41DEL
    IF(T41.LT.T41MAX) GO TO 55
    GO TO (1,2,3),KREAD
400 IF(J.NE.0) GO TO 230
    WRITE (6,2060) T21
    GO TO 180
999 FORMAT (10I1)
1000 FORMAT (5F10.4)
1010 FORMAT(1H1,41X,47HBRAYTON CYCLE - TWO SPOOL - VARIABLE WORK SPLIT)
1021 FORMAT(1H0,40X,49HINTERCOOL ONLY - ALTERNATOR ON LOW PRESSURE SHAF
    1T)
1022 FORMAT(1H0,37X,56HINTERCOOL AND REHEAT - ALTERNATOR ON HIGH PRESSU
    1RE SHAFT)
1023 FORMAT(1H0,37X,55HINTERCOOL AND REHEAT - ALTERNATOR ON LOW PRESSUR
    1E SHAFT)
1024 FORMAT(1H0,36X,57HNO INTERCOOL OR REHEAT - ALTERNATOR ON LOW PRESS
    1URE SHAFT)
1025 FORMAT(1H0,42X,46HREHEAT ONLY - ALTERNATOR ON LOW PRESSURE SHAFT)
1030 FORMAT(1H0,2X,7HETA(T1),3X,7HETA(T2),3X,7HETA(C1),3X,7HETA(C2),5X,
    14HR(L),8X,1HE,7X,5HGAMMA,6X,2HCL,8X,2HC4,5X,8HSPLIT(T)/10F10.4)
1060 FORMAT (1H )
1035 FORMAT(1H0,5X,2HT1,8X,2HTS,9X,3HEPS,7X,2MHR,17X,3HY1P,5X,6HY1PFR6,
    16X,2HY2,7X,5HY2FR6/4F10.2,10X,4F10.4)
1045 FORMAT(1HK,8H T4/T1=,F6.4,5X,8HT4PP/T1=,F6.4,5X,8HT1PP/T1=,F6.4,
    1 //131H
    1 T2/T1 ETACY WCT/P T1P/T1 T3/T1 T4P/T1 T5/T1 T6/T1
    2 P1/P1P P1PP/P2 P4P/P4 P5/P4PP SPLIT(C) ARAD/P AINT/P ATOT/P )
1065 FORMAT(1X,2F8.4,F9.3,10F8.4,3F8.2)
1070 FORMAT(1X,2F8.4,F9.3,10F8.4,3X,12HNOT COMPUTED)
2060 FORMAT(1X,F8.4,10X,17HNO VALID SOLUTION)
    END

```

APPENDIX B

SUBROUTINE RADTRA

Subroutine RADTRA computes the specific prime radiator areas required for both primary heat rejection and intercooling. This subroutine is entered after cycle temperatures, cycle efficiency, and weight flow parameter are computed for each combination of the cycle temperature variables T41 and T21. It is bypassed during the cycle efficiency maximization iteration until the optimum value of T21 is determined. The equations used for calculating specific prime radiator area are those in references 1 and 2.

PROGRAM VARIABLES

The variable names common to both subroutine RADTRA and main program ETACY2 are listed in appendix A and omitted here. The variables for RADTRA are the following:

B	function of TW3
CC	function of EPS and HR
DELTW	iteration correction to TW3
DF	derivative of B with respect to TW3
G	radiator-area equation term
LOOP	iteration counter
P	radiator-area equation term
TW3	wall temperature at radiator inlet
TW9	wall temperature at intercooler inlet
TW10	wall temperature at intercooler exit
TW40	wall temperature at radiator exit
T30	radiator-inlet temperature
T40	radiator-exit temperature
T90	intercooler-inlet temperature
T91	T4P1 in appendix A
T100	intercooler-exit temperature
T101	T4PP1 in appendix A

W radiator-area equation term
WCPDP function of WCPTOP
WCPTOP WPARAM in appendix A

PROGRAM LISTING

\$IBFTC RADTRA

```

SUBROUTINE RADTRA(T1,TS,T31,T41,T91,T101,EPS,HR,WCPTOP,DLMIN,
1AINOP,AROP,ATOP)
  T30 = T31*T1
  T40 = T41*T1
  T90 = T91*T1
  T100 = T101*T1
  WCPDP=60.0*WCPTOP/T1
  TW3 = T30
  LOOP = 0
  CC = .173E-8*EPS/HR
90 B = TW3-T30+CC*(TW3**4-      TS**4)
  DF = 1.0+4.0*CC*TW3**3
  DELTW = B/DF
  IF (ABS(DELTW/TW3)-DLMIN) 91,91,92
92 TW3 = TW3-DELTW
  LOOP = LOOP + 1
  IF (LOOP-75) 90,72,72
91 TW40 = T40
  LOOP = 0
95 B = TW40-T40 +CC*(TW40**4-      TS**4)
  DF = 1.0 +4.0*CC*TW40**3
  DELTW = B/DF
  IF (ABS(DELTW/TW40)-DLMIN) 96,96,97
97 TW40 = TW40 - DELTW
  LOOP = LOOP + 1
  IF (LOOP-75) 95,73,73
96 IF(TS)98,98,99
98 AROP = WCPDP*(240.0*ALOG(TW3/TW40)/HR+20.0*(1.0/TW40**3-1.0
1/TW3**3)/(.173E-8*EPS))
  GO TO 76
99 P = 60.0*ALOG((TW3**4-      TS**4)/(TW40**4-      TS**4))/HR
  W = ALOG((TW3-      TS)*(TW40+      TS)/((TW40-      TS)
1*(TW3+TS)))
  G = 2.0*(ATAN(TW3/TS      )-ATAN(TW40/TS      ))
  AROP = WCPDP*(P+15.0*(W-G)/(.173E-8*EPS*      TS**3))
76 CONTINUE
  GO TO 50
72 AROP = 0.0
  GO TO 50
73 AROP = 0.0
50 TW9 = T90
  LOOP = 0

```

```

54 B = TW9-T90+CC*(TW9**4-      TS**4)
   DF = 1.0+4.0*CC*TW9**3
   DELTW = B/DF
   IF (ABS(DELTW/TW9)-DLMIN) 55,55,56
56 TW9 = TW9-DELTW
   LOOP = LOOP + 1
   IF (LOOP-75) 54,70,70
55 TW10 = T100
   LOOP = 0
57 B = TW10-T100+CC*(TW10**4-      TS**4)
   DF = 1.0 +4.0*CC*TW10**3
   DELTW = B/DF
   IF (ABS(DELTW/TW10)-DLMIN) 59,59,60
60 TW10 = TW10 -DELTW
   LOOP = LOOP + 1
   IF (LOOP-75) 57,71,71
59 IF(TS)62,62,63
62 AINOP = WCPDP*(240.0*ALOG(TW9/TW10)/HR+20.0*(1.0/TW10**3-1.0
  1/TW9**3)/(.173E-8*EPS))
   GO TO 51
63 P = 60.0*ALOG((TW9**4-      TS**4)/(TW10**4-      TS**4))/HR
   W = ALOG((TW9-      TS)*(TW10+      TS)/((TW10-      TS)
  1*(TW9+TS)))
   G = 2.0*(ATAN(TW9/TS      )-ATAN(TW10/TS      ))
   AINOP = WCPDP*(P+15.0*(W-G)/(.173E-8*EPS*      TS**3))
   GO TO 51
70 AINOP = 0.0
   GO TO 51
71 AINOP = 0.0
51 ATOP = AINOP + AROP
   RETURN
   END

```

APPENDIX C

SUBROUTINE MAXIM

Subroutine MAXIM provides the logic for determining the maximum cycle efficiency and associated turbine temperature ratio (T21) within the range specified by T21BEG and T21END. For each cycle temperature ratio, this subroutine is entered after the parametric results for all turbine temperature ratios have been determined. The subroutine logic causes the maximum function value to be approached from both sides until there is a convergence at the peak.

PROGRAM VARIABLES

The variables for MAXIM are as follows:

JUMP	logic indicator for branching
SPEED	retained values of WA
TOLER	convergence tolerance
WA	T21 in appendix A
WAMAX	T21END in appendix A
WEIGHT	retained values of WTFL
WTFL	ETACY in appendix A

PROGRAM LISTING

\$IBFTC MAXIM

```
      SUBROUTINE MAXIM (WA,WAMAX,WTFL,IND,TOLER)
      DIMENSION SPEED(3),WEIGHT(3)
C     IND = 1, FIRST POINT
C     IND = 2, SECOND POINT
C     IND = 3, THIRD POINT (ELIMINATE 1 POINT IF NECESSARY)
C     IND = 4, FOURTH POINT (ELIMINATE 4TH POINT)
C     IND = 5, WAMAX IS LESS THAN WA
      GO TO (140,150,210,270,370),IND
```

```

140 JUMP = 1
    IF (WAMAX.LT.WA) GO TO 145
    SPEED(1) = WA
    WEIGHT(1) = WTFL
    WA = WAMAX
    IND = 2
    RETURN
145 SPEED(3) = WA
    WEIGHT(3) = WTFL
    WA = WAMAX
    IND = 5
    RETURN
150 SPEED(3) = WA
    WEIGHT(3) = WTFL
160 WA = (SPEED(1)+SPEED(3))/2.
    IF((SPEED(3)-SPEED(1)).LT.TOLER) GO TO 400
    IND = 3
    RETURN
210 SPEED(2) = WA
    WEIGHT(2) = WTFL
    IF(WTFL.LE.WEIGHT(1).OR.WTFL.LE.WEIGHT(3)) GO TO 268
    IND = 4

C
C   CHOOSE PROPER INTERVAL FOR NEXT POINT (4TH POINT HAS BEEN ELIMINATED)
C
245 GO TO (246,247),JUMP
246 JUMP = 2
    GO TO 250
247 JUMP = 1
    GO TO 260
250 WA = (SPEED(1)+SPEED(2))/2.0
    RETURN
260 WA = (SPEED(3)+SPEED(2))/2.0
    RETURN
268 IF(WEIGHT(3).GT.WEIGHT(1)) GO TO 269
    WEIGHT(3) = WTFL
    SPEED(3) = WA
    GO TO 160
269 WEIGHT(1) = WTFL
    SPEED(1) = WA
    GO TO 160
270 IF((SPEED(3)-SPEED(1)).LT.TOLER) GO TO 400
280 IF (WTFL-WEIGHT(2)) 320,350,290

C
C   NEW POINT BECOMES MIDPOINT, MIDPOINT BECOMES END POINT
C
290 IF (WA-SPEED(2)) 310,300,300
300 SPEED(1) = SPEED(2)
    SPEED(2) = WA
    WEIGHT(1) = WEIGHT(2)
    WEIGHT(2) = WTFL
    GO TO 245
310 SPEED(3) = SPEED(2)
    SPEED(2) = WA
    WEIGHT(3) = WEIGHT(2)
    WEIGHT(2) = WTFL
    GO TO 245

```

```

C
C   NEW POINT BECOMES END POINT
C
320 IF (WA-SPEED(2)) 340,330,330
330 WEIGHT(3) = WTFL
    SPEED(3) = WA
    GO TO 245
340 WEIGHT(1) = WTFL
    SPEED(1) = WA
    GO TO 245
350 IF(WA.GT.SPEED(2)) GO TO 360
    SPEED(3) = SPEED(2)
    WEIGHT(3) = WEIGHT(2)
    GO TO 210
360 SPEED(1) = SPEED(2)
    WEIGHT(1) = WEIGHT(2)
    GO TO 210
370 SPEED(1) = WA
    WEIGHT(1) = WTFL
    GO TO 160
400 IND = 6
    RETURN
    END

```


APPENDIX D

SUBROUTINE ZERO

Subroutine ZERO provides the logic for determining the turbine work split (ST) that yields a two-shaft one-compressor arrangement ($SC = 0$). When the input value of ST is negative, subroutine ZERO is entered after each calculation of SC in order to test the value of SC and select a new estimation for ST. A linear interpolation or extrapolation is used to determine the estimated value of ST.

PROGRAM VARIABLES

The variables for ZERO are as follows:

TOL	convergence tolerance
X	ST in appendix A
XX	value of ST for second iteration
XY	reciprocal of slope of estimation line
X1	X value of first point defining estimation line
X2	X value of second point defining estimation line
Y	SC in appendix A
Y1	Y value of first point defining estimation line
Y2	Y value of second point defining estimation line

PROGRAM LISTING

\$1BFTC ZERO

```
C
C   FOR MONOTONIC DECREASING Y=FCN(X)
C
SUBROUTINE ZERO (XX,X,Y,IN,TOL)
IF(ABS(Y).LE.TOL) GO TO 100
GO TO (10,20), IN
```

```

10 X1=X
   Y1=Y
   X=XX
   IN=2
   RETURN
20 X2=X
   Y2=Y
   XY=(X2-X1)/(Y2-Y1)
   IF(XY.GE.0.0) GO TO 200
   X=XY*(-Y1)+X1
   IF(ABS(Y1)-ABS(Y2)) 50,40,40
40 X1=X2
   Y1=Y2
50 RETURN
100 IN=3
    Y=ABS(Y)
    RETURN
200 IN=4
    RETURN
    END

```

REFERENCES

1. Glassman, Arthur J.: Thermodynamic and Turbomachinery Concepts For Radioisotope and Reactor Brayton-Cycle Space Power Systems. NASA TN D-2968, 1965.
2. Glassman, Arthur J.: Effect Of Turbine-Coolant Flow On Brayton-Cycle Space-Power System Thermodynamic Performance. NASA TN D-3474, 1966.